BASIC DESIGN PROGRAM FOR RUBBER PROPELLERS

Fred Rash
4614 Mitchell Road
Kingsport, TN 37664

The propeller design papers of Professor E. Eugene Larrabee (Ref. 1-2) aroused considerable interest among modelers, but no computer program has been made widely available. This paper attempts to fill that void. While Prof Larrabee's method has been used, the computer code and any errors are my own. Changes in a few program lines enable design of either outdoor or indoor propellers. This program is written in interpeter BASIC for IBM and Compatible PC's but the dialect is generic enough so that you should be able to run it, with minor changes, on almost any BASIC.

This program provides convenient output to screen and/or printer and converts most common units for you. Entering a question mark usually will give you a menu of acceptable units.

Input of the energy (and turns) delivered by your rubber motor is required. Most interested readers will have your own methods for obtaining this information. If not, please refer to my paper in the 1990 Sympo (Ref. 3). For indoor models, note that landing turns and energy should be subtracted from the launch turns and energy. Charlie Sotich once wrote that the best indoor flights land with about 10% of the launch turns remaining (Ref. 4).

This program first balances average climb angle to average shaft power using an estimated prop efficiency. Later iterations adjust both climb angle and efficiency. Typical average climb angles for outdoor models range from 10 to 30 degrees while indoor models have angles of 0 to 5 degrees. The climb equation is equivalent to Equation 28 of Phillips (Ref. 5-6) but efficiency is handled differently.

Then you will be asked to enter prop diameter, motor run time, and climbing coefficient of lift (Cl). Good data for coefficients of lift and drag for model calculations are difficult to obtain. For outdoor, my initial value for Cl is 0.5. If there is too much power for this Cl, reduce Cl or lengthen motor run. I start with 30 to 90 seconds motor run for conventional Coupe, Wakefield, or Mulvihill models and reduce Cl as little as possible to get a workable average climb angle. A typical climbing Cl may be 0.4-0.5. Models with short motor runs will require lower values for the climbing lift coefficient (Cl), even as low as 0.25.

For outdoor models, the climbing coefficient of drag (Cd) is automatically calculated from the Cl using the equation of Gard (Ref. 7). This Cd (typically 0.05-0.08) should be satisfactory for modern Wakefield, Mulvihill, and Coupe models but should be increased somewhat for fat Oldtimer models.

You will also be given the option to change prop blade lift coefficient (Pl), and prop blade drag coefficient (Pd). I recommend that you not change these until you have tried the default Pl (0.65) and Pd (0.065) values supplied by this program. Then make the miminum number of changes required. Try ranges for Pl of 0.45-0.75, for Pd of 0.05-0.08, and Pl/Pd of 7-10.

When you get a printed table of prop radii, blade angles, and blade widths, examine the prop diameter and blade width. If these seem reasonable, build the prop. If the blades appear too wide, continue the design process. The program will adjust the blade width to maintain the power balance. Continue until you are satisfied with the appearance of the design. This program seems to err toward blades that seem too wide. The propeller may be arbitrarily narrowed. Higher values of PL and PD (keeping the same ratio) are somewhat effective in reducing blade chord and more satisfying.

For indoor, I recommend for simplicity that Cl be based on wing area alone and be kept at 1.0. Drag coefficient (Cd) can then range from about 0.10 to 0.18 (0.13 to 0.18 for Pennyplane down to 0.10 for F1D) but try the default Cd (0.14) first. Prop blade lift coefficient should not require much change from 0.6 (range 0.5-0.7). Instead, vary Pd if required using a range of 0.05-0.12) keeping Pl/Pd in the range of 4 to 8. If you change Cl use only a range of 0.9 to 1.15 with the possible exception of slightly higher Cl for a tandem design. If you fly tandems or very large stabs with aft CG's, then consider a more detailed look at Cl and Cd. Hunt (Ref. 8) and McLean (Ref. 9) have calculated or measured Cl and Cd values for indoor wings, stabs, and entire airframes including tandem and biplane designs. Erbach (Ref. 10) also calculates level flight power requirements and flying speed based on Cl, Cd, and angles of attack of both wing and stab.

If you hesitate to type all of this in, send a letter or call me to get a disk. I retained the copyright notice since I have sold a few copies of this program with other software and want the option to continue. By submission to this NFFS Symposium, I am requesting NFFS to publish and am encouraging you to use, share, and improve this program.

The longer listing below is the outdoor design program. To get an indoor prop design program from it, edit these nine lines and save as PROPMAKR.IND or your preferred name.

```
100 REM BASIC INDOOR PROPELLER DESIGN PROGRAM 3.2 WITH UNITS CONVERTER FOR NFFS SYMPOSIUM 1991 (.I20)

350 CL=1:REM CL IS CLIMBING COEFF. OF LIFT BASED ON WING AREA (OR ON TOTAL AREA IF PREFERRED).

360 CD=.14:REM CD IS COEFF. OF DRAG

370 PL=.6:REM PL IS COEFF. OF LIFT FOR PROP BLADE

380 PD=.08:REM PD IS COEFF. OF DRAG FOR PROP BLADE

1440 P$="AVERAGE POWER AVAILABLE FROM MOTOR IN IN-OZ/SEC ":VAR=WTTS*12*16*.73756:ND=5:GOSUB 1960

1470 REM

3030 D5$="MINUTES":TSC=1/60:REM DEFAULT UNITS FOR TIME FOR INDOOR

3070 D9$="INCH-OUNCES / SECOND":HSC=12*16*.73756:REM DEFAULT UNITS FOR POWER
```

REFERENCES

(1) Larrabee, E. Eugene, "Analytic Designs of Propellers Having Minimum Induced Loss", NFFS Symposium, 1977, pp. 90-98.

(2) Larrabee, E. Eugene, "Propeller Design and Analysis for Modelers", 1979 NFFS International Symposium Report, Bakersfield, CA, pp. 9-25.

(3) Rash, Fred H., "Rubber: Calculating the Capability - BASIC Program for Generic Rubber Motor", NFFS Symposium, 1990, pp. 43-45.

(4) Charlie Sotich, Reference unknown.

(5) W. Hewitt Phillips, "Design of Electrically Powered Models for Maximum Climb," NFFS Symposium 1978, pp. 60-68, Equation 28.

(6) Thanks to W. Hewitt Phillips for references and correspondence and to Donald W. Lane (a local non-modeler and chemical engineer) for discussion and derivations on the climb equation.

(7) Gard, John, "Determination of Flight Trajectory for Rubber Powered Model Aircraft", NFFS Symposium, 1968, pp. 17-27.

(8) Hunt, B. J., "Design and Flight Simulation of Indoor Duration Models Using a Home Microcomputer", Free-Flight Experts' Forum, Organized at the 1986 Model Engineer Exhibition by the Society of Model Aeronautical Engineers, pp. 40-49.

(9) McLean, Doug, "A Method for Predicting Indoor Model Duration", NFFS Symposium, 1976, pp. 54-60.

(10) Erbach, Walter, "Calculating Indoor Flight", NFFS Symposium, 1990, pp. 79-85.

BIOGRAPHY

For 25 years Fred H. Rash has been an industrial organic chemist with Eastman Chemical Company where he has served in several research, development, and production assignments. He received a B.S. degree from Wake Forest University and a Ph.D. from Duke University. He and his wife Judy are parents of a daughter in veterinary school and a son in college. For 23 years he has been the only free flight enthusiast in the local RC club, the Tri-Cities Aeromodelers. He and the other TCA members have flown their strange toys for a few thousand spectators at the annual community festival. A rubber powered hang glider, a pusher flying wing, an 020 powered handkerchief, and a one-bladed 020 helicopter have been his contributions to this air show. He has flown outdoor rubber and gliders for sport and competition but since the US Indoor championships came to neighboring Johnson City, TN, he has been learning indoor. He has also wasted many good flying opportunities staring at a computer screen.

```
100 REM BASIC OUTDOOR PROPELLER DESIGN PROGRAM 3.2 WITH UNITS
CONVERTER FOR NFFS SYMPOSIUM 1991 (.O20)
110 REM COPYRIGHT (C) 1991 BY FRED H. RASH.  RESALE RIGHTS RESERVED.
120 REM LAST REVISED JULY 18, 1991; FIRST VERSION 1983
130 REM LARRABEE PROP DESIGN FROM NFFS SYMPO 1977 PAGES 90-98
140 REM
150 DIM A$(70),B$(70),D$(70),I$(70),P$(70),Q$(70),U$(20)
160 PRINT :PRINT "PROPELLER DESIGN PROGRAM":PRINT
170 PRINT "COPYRIGHT (C) 1991 BY FRED H. RASH.  RESALE RIGHTS
RESERVED.":PRINT :PRINT
180 GOSUB 2210:REM OUTPUT TO SCREEN, PRINTER, OR BOTH
190 GOSUB 2120:REM LABELS FOR DESIGN, DATE, MODELER, NOTES
200 REM NMAX = NUMBER OF INTERVALS ALONG PROP RADIUS
210 REM MAX VALUE FOR NMAX=100 FOR MOST COMPUTERS. TRY 10 OR 20.
220 NMAX=20:REM NMAX MUST BE DIVISIBLE BY 10.
230 DIM XI(NMAX),KI(NMAX),FB(NMAX),G(NMAX)
240 DIM PL(NMAX),PD(NMAX),PHI(NMAX),BETA(NMAX)
250 DIM CHRD(NMAX),STASHUN(NMAX),Y(NMAX)
260 DIM D1$(20),D2$(20),D3$(20),D4$(20),D5$(20),D6$(20),D7$(20), D8$(20),D9$(20)
270 GOSUB 2980:REM UNIT DEFAULTS SET
280 PI=4*ATN(1):REM FINDS PI
290 P$="TOTAL WEIGHT OF MODEL IN ":U$="W":ND=3:GOSUB 2320:
M=VAR/WSC:PRINT
300 M=M/1000:REM IN KG
310 G=9.806:REM ACCELERATION OF GRAVITY IN KG-METER-SEC UNITS
320 P$="PROJECTED WING AREA IN ":U$="R":ND=4:GOSUB 2320:SQIN=
VAR*RSC:PRINT
330 S=.00064516*SQIN:REM SQ METERS
340 RHO=1.225:REM DENSITY OF AIR AT STANDARD SEA LEVEL CONDITIONS
IN KG/M3
350 CL=.5:REM CL IS CLIMBING COEFF. OF LIFT BASED ON WING AREA.   360
REM
370 PL=.65:REM PL IS COEFF. OF LIFT FOR PROP BLADE
380 PD=.065:REM PD IS COEFF. OF DRAG FOR PROP BLADE
390 REM ENERGY INPUTS
400 P$="ENERGY OF RUBBER MOTOR FROM RUBMOTOR PROGRAM IN
":U$="E":ND=4: GOSUB 2320:ENERGY=VAR/ESC:PRINT
410 PRINT "ENTER ACTUAL TURNS IN RUBBER MOTOR FROM RUBMOTOR
PROGRAM": PRINT :INPUT TURNS
420 P$="ACTUAL TURNS IN RUBBER MOTOR ":VAR=TURNS:ND=1:GOSUB
1960
430 EFF=.75:REM INITIAL ESTIMATE FOR PROP EFFICIENCY
440 GOSUB 1370:REM CHANGE INPUTS
450 GOSUB 1560:REM BALANCE CLIMB ANGLE TO AVERAGE AVAILABLE
POWER
460 GOSUB 590:REM DESIGN PROP
470 GOSUB 1560:REM REBALANCE CLIMB ANGLE TO AVERAGE POWER
480 GOSUB 590:REM REDESIGN PROP
490 GOSUB 1560:REM REBALANCE CLIMB ANGLE TO AVERAGE POWER
500 GOSUB 590:REM REDESIGN PROP
510 GOSUB 1730:REM PRINT THRUST & POWER COEFFICIENTS
520 GOSUB 970:REM PRINT DESIGN AND EVALUATION
530 IF WU/WTTS>1.02 THEN PRINT "REDUCE PROP DIAMETER A LITTLE, OR
SHORTEN MOTOR RUN":GOSUB 1370:GOTO 490
```

```
540 IF WU/WTTS<.98 THEN PRINT "REDUCE CL, LENGTHEN MOTOR RUN, OR
INCREASE PROP DIA":GOSUB 1370:GOTO 490
550 IF WU/WTTS>.98 AND WU/WTTS<1.02 THEN P$="POWER BALANCE IS
GOOD.":GOSUB 2050:P$="":GOSUB 2050
560 PRINT "DO YOU WANT TO CONTINUE THIS DESIGN (Y/N)?":INPUT A$:IF
A$="Y" OR A$="y" THEN GOSUB 1370:GOTO 490
570 END
580 REM DESIGN PROP
590 LAMDA=V/(2*PI*RPS*R):REM ADVANCE RATIO
600 J=PI*LAMDA
610 B=2:REM NUMBER OF BLADES
620 FOR N=0 TO NMAX
630 XI(N)=N/NMAX
640 KI(N)=XI(N)/LAMDA
650 X=EXP(-(B/2)*SQR(1+LAMDA*LAMDA)/LAMDA*(1-XI(N)))
660 FB(N)=(2/PI)*(ATN(SQR(1-X*X)/(X+(1-SGN(X)*SGN(X))))+PI/4* ((1-
SGN(X))*(1-SGN(X)))):REM ARCCOS
670 G(N)=FB(N)*(KI(N)*KI(N)/(1+KI(N)*KI(N)))
680 Y(N)=G(N)*XI(N):REM GXI(N)
690 NEXT N
700 XA=0:XB=1:GOSUB 1800:REM CALLS SIMPSON'S RULE INTEGRATION
710 VR=TC/NGL/4
720 P$="VRATIO":VAR=VR:ND=5:GOSUB 1960
730 FOR COLUMN=1 TO 3
740 FOR N=0 TO NMAX:Y(N)=(1+G(N)*VR/2)*G(N)*VR*XI(N):NEXT N
750 XA=0:XB=1:GOSUB 1800:REM CALLS SIMPSON'S RULE INTEGRATION
760 TN=4*NGL:VR=TC/TN*VR
770 NEXT COLUMN
780 P$="NEW VRATIO ":VAR=VR:ND=5:GOSUB 1960
790 FOR N=0 TO NMAX
800 PL(N)=PL
810 PD(N)=PD
820 ALPHA=.1:REM 0.1 RADIAN IS 5.73 DEGREES
830 IF N=0 THEN PHI(N)=0:BETA(N)=PI/2:GOTO 860
840 PHI(N)=ATN((LAMDA*NMAX/N)*(1+VR/2))
850 BETA(N)=ALPHA+PHI(N)
860 CHRD(N)=VR*R/PL(N)*(4*PI/B)*(LAMDA/SQR(1+KI(N)*KI(N)))*G(N)
870 STASHUN(N)=XI(N)*R
880 IF ABS(N-.7*NMAX)<.000001 THEN
EPSIL=ATN(PD(N)/PL(N)):PHI=PHI(N):P=2*PI*STASHUN(N)*SIN(BETA(N))/COS
(BETA(N))
890 NEXT N
900 REM CALCULATE PROP EFFICIENCY & TORQUE USING 0.7 RADIUS POINT
910 EFF=(SIN(PHI))*(COS(PHI+EPSIL))/(SIN(PHI+EPSIL))/(COS(PHI))/ (1+VR/2)
920 Q=V*TN*(RHO*V*V/2)*(PI*R*R)/(2*PI*RPS)/EFF
930 T=TN*RHO*V*V*PI*R*R/2:REM T IS THRUST
940 PN=TN/EFF
950 WU=PN*RHO*V*V*V*PI*R*R/2:REM WATTS OF POWER CONSUMED BY
PROP
960 RETURN
970 PRINT :PRINT "ENTER IN OR MM FOR UNITS TO PRINT PROP BLOCK
LAYOUT"
980 INPUT A$
990 IF A$="MM" THEN SC=1000
```

```
1000 IF A$="IN" THEN SC=39.37
1010 IF A$<>"IN" AND A$<>"MM" THEN GOTO 970
1020 FOR A=A0 TO A1
1030 IF A1=1 AND A=A1 THEN LPRINT :LPRINT ,"XI","STATION",
"BETA","CHORD"
1040 IF A0=0 AND A=A0 THEN PRINT :PRINT ;"XI","STATION", "BETA","CHORD"
1050 IF A1=1 AND A=A1 THEN LPRINT ,"RADIUS",A$,"DEGREES",A$: LPRINT
1060 IF A0=0 AND A=A0 THEN PRINT ;"RADIUS",A$,"DEGREES",A$:PRINT
1070 NEXT A
1080 FOR N=0 TO NMAX
1090 VAR=XI(N):ND=2:GOSUB 1930:XI(N)=VAR
1100 VAR=CHRD(N)*SC:ND=2:GOSUB 1930:CHRD(N)=VAR
1110 VAR=BETA(N)*180/PI:ND=1:GOSUB 1930:BETA(N)=VAR
1120 VAR=STASHUN(N)*SC:ND=2:GOSUB 1930:STASHUN(N)=VAR
1130 FOR A=A0 TO A1
1140 IF A1=1 AND A=A1 THEN LPRINT ,XI(N),STASHUN(N), BETA(N),CHRD(N)
1150 IF A0=0 AND A=A0 THEN PRINT ;XI(N),STASHUN(N), BETA(N),CHRD(N)
1160 NEXT A
1170 NEXT N
1180 P$="":GOSUB 2050:GOSUB 2050
1190 REM OUTPUT PROP EFFICIENCY & TORQUE
1200 P$="PROP DIAMETER IN ":U$="L":VAR=DIAMETER:ND=3:GOSUB 3190
1210 VAR=VAR*LSC:GOSUB 3270:GOSUB 1960
1220 P$="PITCH (GEOMETRIC) IN ":Q$=D1$:GOSUB
3270:VAR=P*39.37*LSC:ND=3:GOSUB 1960
1230 P$="(GEOMETRIC) PITCH/DIAMETER RATIO":VAR=P/(2*R):ND=3:GOSUB
1960
1240 P$="EFFECTIVE PITCH/DIAMETER RATIO":VAR=J:ND=3:GOSUB 1960
1250 P$="ADVANCE RATIO LAMDA":VAR=LAMDA:ND=5:GOSUB 1960
1260 P$="THRUST IN ":U$="F":VAR=T:ND=6:GOSUB 3190
1270 VAR=VAR*FSC:GOSUB 3270:GOSUB 1960
1280 P$="AIR DENSITY (RHO) IN KG/M3":VAR=RHO:ND=3:GOSUB 1960
1290 P$="VRATIO":VAR=VR:ND=5:GOSUB 1960
1300 P$="EFFICIENCY":VAR=EFF:ND=3:GOSUB 1960
1310 P$="TORQUE IN ":U$="Q":VAR=Q:ND=6:GOSUB 3190
1320 VAR=VAR*QSC:GOSUB 3270:GOSUB 1960
1330 P$="POWER USED BY PROP IN ":U$="H":VAR=WU:ND=5:GOSUB 3190
1340 VAR=VAR*HSC:GOSUB 3270:GOSUB 1960
1350 P$="AVERAGE POWER AVAILABLE FROM MOTOR IN ":Q$=D9$:GOSUB
3270:VAR=WTTS*HSC:ND=5:GOSUB 1960
1360 RETURN
1370 P$="":GOSUB 2050:REM MODIFY CERTAIN INPUTS HERE
1380 P$="PROP DIAMETER IN ":U$="L":ND=3:GOSUB 2320:DIAMETER=
VAR/LSC:PRINT
1390 R=DIAMETER/2/39.37:REM R IN METERS
1400 P$="MOTOR RUN TIME IN ":U$="T":ND=2:GOSUB 2320:TIME=VAR/TSC
1410 RPS=TURNS/TIME
1420 P$="PROP ROTATIONAL RATE IN REVS PER SECOND ":VAR=RPS:ND=2:
GOSUB 1960
1430 WTTS=ENERGY/TIME*1.35582:REM CALCULATE POWER
1440 P$="AVERAGE POWER AVAILABLE FROM MOTOR IN FT-LB/SEC ":VAR=
WTTS*.73756:ND=5:GOSUB 1960
1450 P$="":GOSUB 2050
```

```
1460 P$="CLIMBING COEFFICIENT OF LIFT CL":VAR=CL:ND=2:GOSUB
3280:CL=VAR
1470 CD=.05+.0405*CL*CL*CL:REM CD IS CLIMBING COEFF. OF DRAG BASED
ON SAME AREA AS CL.  SEE GARD.
1480 P$="CLIMBING COEFFICIENT OF DRAG CD":VAR=CD:ND=3:GOSUB
3280:CD=VAR
1490 P$="":GOSUB 2050
1500 P$="COEFFICIENT OF LIFT PL FOR PROP BLADES":VAR=PL:ND=2:
GOSUB 3280:PL=VAR
1510 P$="COEFFICIENT OF DRAG PD FOR PROP BLADES":VAR=PD:ND=3:
GOSUB 3280:PD=VAR
1520 P$="":GOSUB 2050
1530 RETURN
1540 REM BALANCE AVERAGE CLIMB ANGLE TO AVERAGE AVAILABLE
POWER
1550 REM EFF*WTTS/(M*G)*SQR((RHO*S*CL)/(2*M*G)) =
SQR(COS(GAMMA))*(CD/CL*COS(GAMMA)+SIN(GAMMA))
1560 PF=EFF*WTTS/(M*G)*SQR((RHO*S*CL)/(2*M*G)):REM AVAILABLE
POWER FACTOR
1570 GMAX=ATN((SQR(9*(CD/CL)*(CD/CL)+8)-3*(CD/CL))/2):REM DERIVATIVE
USED TO FIND MAX VALUE OF CLIMB ANGLE GAMMA
1580 PMAX=SQR(COS(GMAX))*(CD/CL*COS(GMAX)+SIN(GMAX)):REM
POWER FOR MAX CLIMB ANGLE
1590 IF PF>PMAX THEN PRINT "REDUCE CL OR INCREASE MOTOR RUN
TIME":PRINT :GOSUB 1400:GOTO 1560
1600 GAMMA=PF:REM FIRST GUESS FOR ITERATION
1610 E=PF-SQR(COS(GAMMA))*(CD/CL*COS(GAMMA)+SIN(GAMMA))
1620 GAMMA=GAMMA+E
1630 IF GAMMA>GMAX THEN GAMMA=GAMMA-2*E
1640 IF ABS(E)>.000001 THEN GOTO 1610
1650 P$="":GOSUB 2050
1660 P$="AVERAGE CLIMB ANGLE IN DEGREES
":VAR=GAMMA*180/PI:ND=1:GOSUB 1960
1670 V=SQR((2*M*G*COS(GAMMA))/(RHO*S*CL))
1680 REM TC IS THRUST COEFFICIENT
1690 TC=CD*S/(PI*R*R)+(M*G*SIN(GAMMA))/(.5*RHO*V*V*PI*R*R)
1700 REM CALCULATE POWER COEFFICIENT
1710 PC=2*WTTS/(RHO*V*V*V*PI*R*R):REM REF LARRABEE BAKERSFIELD
1720 RETURN
1730 REM THRUST & POWER COEFFICIENTS
1740 P$="":GOSUB 2050
1750 P$="VELOCITY IN ":U$="V":VAR=V:ND=3:GOSUB 3190
1760 VAR=VAR*VSC:GOSUB 3270:GOSUB 1960
1770 P$="THRUST COEFFICIENT ":VAR=TC:ND=5:GOSUB 1960
1780 P$="POWER COEFFICIENT ":VAR=PC:ND=5:GOSUB 1960
1790 RETURN
1800 REM SIMPSON'S RULE INTEGRATION
1810 REM XA IS LOWER LIMIT; XB IS UPPER LIMIT
1820 REM Y(N) IS VALUE OF FUNCTION FOR N SUBINTERVALS (FOR N+1
VALUES)
1830 REM N MUST BE EVEN
1840 REM LET WIDTH OF SUBINTERVAL=(XB-XA)/NMAX
1850 NGL=0
1860 FOR N=0 TO NMAX
```

```
1870 IF N=0 THEN NGL=NGL+Y(N):GOTO 1910
1880 IF N=NMAX THEN NGL=NGL+Y(N):GOTO 1910
1890 IF (N/2-INT(N/2))=0 THEN NGL=NGL+2*Y(N):GOTO 1910
1900 NGL=NGL+4*Y(N)
1910 NEXT N
1920 NGL=(XB-XA)/NMAX*NGL/3:RETURN
1930 REM ROUNDER--VARIABLE INPUT AS VAR; NUMBER OF DECIMAL
DIGITS AS ND
1940 POWER=.1:FOR ZZ=0 TO ND:POWER=10*POWER:NEXT ZZ
1950 VAR=INT(POWER*VAR+.5)/POWER:RETURN
1960 REM PRINTS STRING, EQUALS SIGN, BUT NO LINEFEED
1970 REM THEN PRINTS ROUNDED VARIABLE
1980 REM STRING INPUT AS P$; VARIABLE INPUT AS VAR
1990 REM NUMBER OF DECIMAL DIGITS AS ND
2000 GOSUB 1930:REM ROUNDER
2010 FOR A=A0 TO A1
2020 IF A0=0 AND A=A0 THEN PRINT P$;" = ";VAR
2030 IF A1=1 AND A=A1 THEN LPRINT ,P$;" = ";VAR
2040 NEXT A:RETURN
2050 REM PRINT STRING ROUTINE--P$ IS INPUT TO BE PRINTED
2060 REM A0=0 FOR SCREEN; A1=1 FOR PRINTER
2070 FOR A=A0 TO A1
2080 IF A0=0 AND A=A0 THEN PRINT P$
2090 IF A1=1 AND A=A1 THEN LPRINT ,P$
2100 REM LEAVES LEFT MARGIN ON PAPER OUTPUT
2110 NEXT A:RETURN
2120 REM LABELS EACH RUN OF PROGRAM TO REDUCE LATER CONFUSION
2130 PRINT "ENTER NAME OF MODEL DESIGN":PRINT :INPUT A$
2140 P$=A$:GOSUB 2050
2150 PRINT "ENTER DATE":PRINT :INPUT A$
2160 P$=A$:GOSUB 2050
2170 PRINT "ENTER MODELER'S NAME ":PRINT :INPUT A$
2180 P$=A$:GOSUB 2050
2190 PRINT "ENTER OTHER NOTES, IF ANY":PRINT :INPUT A$
2200 P$=A$:GOSUB 2050:P$="":GOSUB 2050:RETURN
2210 REM OUTPUT
2220 PRINT "DO YOU WANT PRINTED OUTPUT TO GO TO"
2230 PRINT "    (S) SCREEN ONLY?"
2240 PRINT "    (P) PRINTER ONLY?"
2250 PRINT "    (B) BOTH SCREEN AND PRINTER?"
2260 PRINT "    ENTER S, P, OR B"
2270 INPUT A$
2280 IF A$="S" OR A$="s" THEN A0=0:A1=0
2290 IF A$="P" OR A$="p" THEN A0=1:A1=1
2300 IF A$="B" OR A$="b" THEN A0=0:A1=1
2310 RETURN
2320 REM INPUT SUBROUTINE TO PERMIT ENTRY IN DIFFERENT UNITS
2330 REM USE ? TO CHANGE UNITS
2340 GOSUB 3090:REM GETS NAME OF UNITS
2350 PRINT "ENTER ";P$;Q$;" OR ENTER ? TO CHANGE UNITS OF ";D$
2360 INPUT I$
2370 IF I$="?" THEN GOSUB 2410:GOTO 2340
2380 VAR=VAL(I$):REM VAR FOR PRINTING, BUT SCALED FOR INTERNAL
USE
```

```
2390 GOSUB 3270:GOSUB 1960:REM PRINT
2400 RETURN
2410 REM CHANGE UNITS SUBROUTINE
2420 REM LENGTH, WEIGHT OR MASS, TORQUE, ENERGY, TIME, THRUST,
VELOCITY, AREA, POWER
2430 RESTORE
2440 READ A$,B$,Q$,SCA
2450 IF A$=U$ THEN PRINT "ENTER ";B$;" FOR ";Q$
2460 IF A$<>"X" THEN GOTO 2440
2470 INPUT I$
2480 RESTORE
2490 READ A$,B$,Q$,SCA
2500 IF A$="X" THEN PRINT "ERROR, TRY AGAIN":GOTO 2470
2510 IF B$<>I$ THEN GOTO 2490
2520 IF A$="L" THEN D1$=Q$:LSC=SCA
2530 IF A$="W" THEN D2$=Q$:WSC=SCA
2540 IF A$="Q" THEN D3$=Q$:QSC=SCA
2550 IF A$="E" THEN D4$=Q$:ESC=SCA
2560 IF A$="T" THEN D5$=Q$:TSC=SCA
2570 IF A$="F" THEN D6$=Q$:FSC=SCA
2580 IF A$="V" THEN D7$=Q$:VSC=SCA
2590 IF A$="R" THEN D8$=Q$:RSC=SCA
2600 IF A$="H" THEN D9$=Q$:HSC=SCA
2610 RETURN
2620 REM DATA TABLE FOR UNITS
2630 DATA L,CM,CENTIMETERS,2.54
2640 DATA L,MM,MILLIMETERS,25.4
2650 DATA L,M,METERS,0.0254
2660 DATA L,IN,INCHES,1
2670 DATA L,FT,FEET,0.08333
2680 DATA W,OZ,OUNCES,0.035273
2690 DATA W,LB,POUNDS,0.0022045
2700 DATA W,G,GRAMS,1
2710 DATA W,KG,KILOGRAMS,0.001
2720 DATA Q,LB-FT,POUND-FEET,0.73756
2730 DATA Q,OZ-IN,OUNCE-INCHES,141.612
2740 DATA Q,G-CM,GRAM-CENTIMETERS,10197.3
2750 DATA Q,KG-CM,KILOGRAM-CENTIMETERS,10.1973
2760 DATA E,FT-LB,FOOT-POUNDS,1
2770 DATA E,IN-OZ,INCH-OUNCES,192
2780 DATA E,J,JOULES,1.3558
2790 DATA E,MJ,MILLIJOULES,1355.8
2800 DATA T,SEC,SECONDS,1
2810 DATA T,MIN,MINUTES,0.016667
2820 DATA F,OZ,OUNCES,3.5968
2830 DATA F,LB,POUNDS,0.2248
2840 DATA F,N,NEWTONS,1
2850 DATA V,FPS,FEET PER SECOND,3.28083
2860 DATA V,MPS,METERS PER SECOND,1
2870 DATA V,MPH,MILES PER HOUR,2.23693
2880 DATA V,KPH,KILOMETERS PER HOUR,3.6
2890 DATA R,SQ IN,SQUARE INCHES,1
2900 DATA R,SQ FT,SQUARE FEET,144
2910 DATA R,SQ DM,SQUARE DECIMETERS,15.5
```

```
2920 DATA R,SQ M,SQUARE METERS,1550
2930 DATA H,W,WATTS,1
2940 DATA H,IN-OZ/SEC,INCH-OUNCES / SECOND,141.61
2950 DATA H,FT-LB/SEC,FOOT-POUNDS / SECOND,0.73756
2960 DATA H,HP,HORSEPOWER,1.341E-03
2970 DATA X,XX,EXIT,0
2980 REM SET DEFAULT UNITS HERE
2990 D1$="INCHES":LSC=1:REM DEFAULT UNITS FOR LENGTH
3000 D2$="GRAMS":WSC=1:REM DEFAULT UNITS FOR WEIGHT OR MASS
3010 D3$="OZ-IN":QSC=141.612:REM DEFAULT UNITS FOR TORQUE
3020 D4$="FT-LB":ESC=1:REM DEFAULT UNITS FOR TOTAL ENERGY
3030 D5$="SECONDS":TSC=1:REM DEFAULT TIME FOR OUTDOOR
3040 D6$="OUNCES":FSC=3.5968:REM DEFAULT UNITS FOR THRUST
3050 D7$="FEET PER SECOND":VSC=3.28083:REM DEFAULT VELOCITY
3060 D8$="SQUARE INCHES":RSC=1:REM DEFAULT UNITS FOR AREA
3070 D9$="FOOT-POUNDS / SECOND":HSC=.73756:REM DEFAULT POWER
3080 RETURN
3090 IF U$="L" THEN Q$=D1$:D$="LENGTH"
3100 IF U$="W" THEN Q$=D2$:D$="WEIGHT"
3110 IF U$="Q" THEN Q$=D3$:D$="TORQUE"
3120 IF U$="E" THEN Q$=D4$:D$="ENERGY"
3130 IF U$="T" THEN Q$=D5$:D$="TIME"
3140 IF U$="F" THEN Q$=D6$:D$="THRUST"
3150 IF U$="V" THEN Q$=D7$:D$="VELOCITY"
3160 IF U$="R" THEN Q$=D8$:D$="AREA"
3170 IF U$="H" THEN Q$=D9$:D$="POWER"
3180 RETURN
3190 REM PRINTS QUANTITY IN DIFFERENT UNITS
3200 REM INPUTS ARE P$,Q$,D$,VAR,ND
3210 GOSUB 3090:REM RESET UNITS AND QUANTITIES
3220 PRINT "PRESS RETURN TO PRINT ";P$;Q$;" OR ENTER ? TO CHANGE
UNITS OF ";D$
3230 INPUT I$
3240 IF I$<>"" AND I$<>"?" THEN GOTO 3220
3250 IF I$="?" THEN GOSUB 2410:GOTO 3210
3260 IF I$="" THEN RETURN
3270 P$=P$+Q$:RETURN
3280 REM PRINT OLD VALUE, ENTER NEW VALUE OR RETURN TO KEEP
SAME VALUE, AND PRINT FINAL VALUE
3290 REM INPUTS ARE P$, VAR, ND
3300 REM OUTPUTS ARE NEW VALUE FOR VAR & MESSAGES TO SCREEN &
PRINTER
3310 GOSUB 1960
3320 PRINT "ENTER NEW ";P$;" OR RETURN TO KEEP SAME VALUE"
3330 INPUT I$:IF I$<>"" THEN VAR=VAL(I$)
3340 GOSUB 1960
3350 P$="":GOSUB 2050:RETURN
3360 END
```